

1 Liste

1.1 Zadatak: Definiranje liste

Liste mogu sadržavati bilo kakve elemente. Primjeri lista

```
(%i1) lista: [2, 3, a, 4];  
      lista2: [2, 15, 3, 15];  
      lista3: [a, b, c];
```

```
(%o1) [2, 3, a, 4]
```

```
(%o2) [2, 15, 3, 15]
```

```
(%o3) [a, b, c]
```

1.2 Zadatak: Neke opcije s listama

1. Pristupanje n-tom elemntu liste: ImeListe[n]

```
(%i4) lista[2];  
      lista3[3];
```

```
(%o4) 3
```

```
(%o5) c
```

2. Pristupanje prvom elemntu liste: ImeListe[1] ili first(ImeListe)

```
(%i6) lista[1];  
      first(lista);
```

```
(%o6) 2
```

```
(%o7) 2
```

3. Sortiranje: sort(ImeListe)

```
(%i8) sort(lista);  
      sort(lista2);
```

```
(%o8) [2, 3, 4, a]
```

```
(%o9) [2, 3, 15, 15]
```

4. Funkcija member Nalazi li se "21" u prvoj, a "c" u trećoj listi?

```
(%i10) member(21, lista);  
       member(c, lista3);
```

```
(%o10) false
```

```
(%o11) true
```

5. Zamjena elemenata u listi U prvoj listi element "2" zamjenite s "22".

```
(%i12) lista[1]: 22$
      lista;
```

```
(%o13) [22, 3, a, 4]
```

U prvoj listi podlistu [3, a] zamjenite s listom [x, y, z].

```
(%i14) lista: append(rest(lista, -3), [x, y, z], rest(lista, 3));
```

```
(%o14) [22, x, y, z, 4]
```

Funkcija `rest(lista, n)` vraća listu bez prvih n elemenata ako je $n > 0$, odnosno bez zadnjih $-n$ elemenata ako je $n < 0$. Broj elemenata liste vraća funkcija `length`

```
(%i15) length(lista);
```

```
(%o15) 5
```

1.3 Zadatak: Definiranje liste pomoću naredbe `make_list`

```
(%i16) l: makelist(i^2, i, 1, 5);
```

```
(%o16) [1, 4, 9, 16, 25]
```

`l` je lista kvadrata prirodnih brojeva od 1 do 5.

1.4 Zadatak: Spajanje lista

Proizvoljan broj lista možete spojiti u jednu. Naredbom `append` liste se spajaju na način da se elementi svake liste nastavljaaju na elemente prethodne liste.

```
(%i17) append(lista, lista2, lista3);
```

```
(%o17) [22, x, y, z, 4, 2, 15, 3, 15, a, b, c]
```

Kučni uradak: Proučite `help` za naredbe `join`, `union`, `append` i `intersection`.

2 Višedimenzionalne liste

Elementi liste mogu biti i liste. Tada govorimo o višedimenzionalnim listama.

2.1 Zadatak: Primjer višedimenzionalne liste

```
(%i18) tocke: makelist([x, sin(x)], x, 0, 2*%pi, 0.5);
```

```
(%o18) [[0, 0], [0.5, 0.479425538604203], [1.0, .8414709848078965],
[1.5, .9974949866040544], [2.0, .9092974268256817], [2.5, .5984721441039565],
```

```
[3.0, .1411200080598672], [3.5, -.3507832276896198], [4.0, -.7568024953079282],  
[4.5, -0.977530117665097], [5.0, -.9589242746631385], [5.5, -.7055403255703919],  
[6.0, -.2794154981989259]]
```

Dimenziju liste možemo i provjeriti.

```
(%i19) [length(tocke), length(tocke[1])];
```

```
(%o19) [13, 2]
```

Dakle tocke je lista od 13 elemenata, od kojih je svaki lista s 2 elementa. Drugom elementu ove liste pristupamo kao i ranije:

```
(%i20) tocke[2];
```

```
(%o20) [0.5, 0.479425538604203]
```

Ukoliko želimo pristupiti prvom elementu ove liste pišemo:

```
(%i21) tocke[2][1];
```


```
(%o21) 0.5
```

Slično i za ostale elemente.


2.2 Zadatak: Crtanje diskretno zadanih grafova

Maxima može i grafički prikazivati liste dimenzije 1 i 2 funkcijama `plot2d` i `wxplot2d`. Prvo pogledajte help za ove naredbe, a zatim prikažite grafički listu tocke.

```
(%i22) wxplot2d([discrete, tocke],[style, points])$  
wxplot2d([discrete, tocke],[style, linespoints])$
```



```
(%t22)
```



(%t23)

```
(%i24) kill(lista, lista2, lista3, lista4, tocke);
```

(%o24) done

3 Matrice

I matrice su liste! Matricu u Maximi upisujemo pomoću funkcije matrix kojoj zadajemo retke (svaki kao jednu listu elemenata).

3.1 Zadatak: Zbrajanje matrica

Neka su zadane dvije matrice:

```
(%i25) mat1: matrix([1, 2], [3, 1]);
        mat2: matrix([0, 1], [2, -2]);
```

(%o25) $\begin{pmatrix} 1 & 2 \\ 3 & 1 \end{pmatrix}$

(%o26) $\begin{pmatrix} 0 & 1 \\ 2 & -2 \end{pmatrix}$

Njihov zbroj je matrica

```
(%i27) zbroj = mat1 + mat2;
```

(%o27) $zbroj = \begin{pmatrix} 1 & 3 \\ 5 & -1 \end{pmatrix}$

3.2 Zadatak: Množenje matrica

Matrice se NE množe znakom '*':

```
(%i28) mat1 * mat2;
```

```
(%o28)  $\begin{pmatrix} 0 & 2 \\ 6 & -2 \end{pmatrix}$ 
```

mat1*mat2 je matrica čiji su elementi umnošci elemenata (na istim pozicijama) matrica mat1 i mat2 (tzv. Hadamardov produkt matrica). Matrice se množe točkom:

```
(%i29) mat1 . mat2;
```

```
(%o29)  $\begin{pmatrix} 4 & -3 \\ 2 & 1 \end{pmatrix}$ 
```

Unešene matrice više ne trebamo:

```
(%i30) kill(mat1, mat2);
```

```
(%o30) done
```

3.3 Zadatak: Neke matrice funkcije

Nađite determinantu, trag, inverz, transponiranu i konjugirano transponiranu matricu neke proizvoljne matrice. Funkcija mattrace za računanje traga matrice nalazi se u paketu "nchrpl" kojeg treba posebno učitati.

```
(%i31) load("nchrpl");  
      mat: matrix([1, %i], [0, 1]);  
      determinant(mat);  
      mattrace(mat);  
      invert(mat);  
      transpose(mat);  
      conjugate(transpose(mat));
```

```
(%o31) /usr/share/maxima/5.23.2/share/matrix/nchrpl.mac
```

```
(%o32)  $\begin{pmatrix} 1 & i \\ 0 & 1 \end{pmatrix}$ 
```

```
(%o33) 1
```

```
(%o34) 2
```

```
(%o35)  $\begin{pmatrix} 1 & -i \\ 0 & 1 \end{pmatrix}$ 
```

```
(%o36)  $\begin{pmatrix} 1 & 0 \\ i & 1 \end{pmatrix}$ 
```

```
(%o37)  $\begin{pmatrix} 1 & 0 \\ -i & 1 \end{pmatrix}$ 
```

Popis i opis funkcija vezanih uz matrice i linearnu algebru možete pronaći ovdje:

http://www.delorie.com/gnu/docs/maxima/maxima_82.html

http://maxima.sourceforge.net/docs/manual/en/maxima_99.html

4 Jednadžbe

Jednadžbe rješavamo funkcijom `solve`. Za početak proučite `help` za ovu funkciju.

4.1 Primjer: Rješavanje jednadžbe

Rješavamo jednadžbu $2x-3=4$, gdje je x nepoznanica.

```
(%i38) solve(2*x - 3 = 4, x);
```

```
(%o38) [x =  $\frac{7}{2}$ ]
```

```
(%i39) solve(2*x - 3 = 4, x), numer;
```

```
rat : replaced - 3.5 by - 7/2 = -3.5
```

```
(%o39) [x = 3.5]
```

4.2 Zadatak: Nultočke i kritične točke

Nađite nultočke i kritične točke (nultočke prve derivacije) funkcije $f(x) = (1 + 5x - 3x^2)(x^2 + x - 2)$.

```
(%i40) f(x) := (1 + 5*x - 3*x^2) * (x^2 + x - 2);  
sol: solve(f(x) = 0, x);  
nultočke: map(rhs, sol);  
nultočke, numer;
```

```
(%o40) f(x) := (1 + 5x + (-3)x^2)(x^2 + x - 2)
```

```
(%o41) [x =  $-\frac{\sqrt{37}-5}{6}$ , x =  $\frac{\sqrt{37}+5}{6}$ , x = 1, x = -2]
```

```
(%o42) [ $-\frac{\sqrt{37}-5}{6}$ ,  $\frac{\sqrt{37}+5}{6}$ , 1, -2]
```

```
(%o43) [-.1804604217163699, 1.847127088383036, 1, -2]
```

Objašnjenje: Funkcija `map(a, b)` primjenjuje `a` na svaki element liste `b`, tj. vraća listu `[a(b[1]), a(b[2]), ...]`. Funkcija `rhs` (znači right-hand side) vraća onaj dio argumenta koji je desno od znaka jednakosti, tj. `rhs(x=y)` vraća `y`.

```
(%i44) f1(x) := expand(diff(f(x), x));  
f1(x);  
solve(f1(x) = 0, x);  
kriticne: map(rhs, %);  
kriticne, numer;
```

```
(%o44) f1(x) := expand(diff(f(x), x))
```

```
(%o45)  $-12x^3 + 6x^2 + 24x - 9$ 
```

$$(\%o46) \left[x = -\frac{\sqrt{3}+1}{2}, x = \frac{\sqrt{3}-1}{2}, x = \frac{3}{2} \right]$$

$$(\%o47) \left[-\frac{\sqrt{3}+1}{2}, \frac{\sqrt{3}-1}{2}, \frac{3}{2} \right]$$

$$(\%o48) [-1.366025403784439, .3660254037844386, 1.5]$$


4.3 Zadatak: Grafički prikaz

Grafički prikažite vrijednosti funkcije iz prethodnog zadatka u kritičnim točkama.

```
(%i49) KT: makelist([kriticne[i], f(kriticne[i])], i, 1, length(kriticne));
      KT, numer;
      wxplot2d([f(x), [discrete, KT]], [x, -2, 2], [style, [lines, 3], [points, 3]], [col
```

$$(\%o49) \left[\left[-\frac{\sqrt{3}+1}{2}, \left(-\frac{3(\sqrt{3}+1)^2}{4} - \frac{5(\sqrt{3}+1)}{2} + 1 \right) \left(\frac{(\sqrt{3}+1)^2}{4} - \frac{\sqrt{3}+1}{2} - 2 \right) \right], \right. \\ \left. \left[\frac{\sqrt{3}-1}{2}, \left(\frac{\sqrt{3}-1}{2} + \frac{(\sqrt{3}-1)^2}{4} - 2 \right) \left(\frac{5(\sqrt{3}-1)}{2} - \frac{3(\sqrt{3}-1)^2}{4} + 1 \right) \right], \right. \\ \left. \left[\frac{3}{2}, \frac{49}{16} \right] \right]$$

$$(\%o50) [[-1.366025403784439, 17.14230484541326], \\ [.3660254037844386, -3.642304845413263], [1.5, 3.0625]]$$



4.4 Zadatak: Sustav jednačbi

Riješite sustav jednačbi $x^2 + y - 2 = 0$, $x - 2y + 1 = 0$.

```
(%i52) solve([x^2 + y - 2 = 0, x - 2*y + 1 = 0], [x, y]);
      fullmapl(rhs, %);
```

$$(\%o52) [[x = 1, y = 1], [x = -\frac{3}{2}, y = -\frac{1}{4}]]$$

$$(\%o53) [[1, 1], [-\frac{3}{2}, -\frac{1}{4}]]$$

4.5 Zadatak: Algebarska kvadratna jednačba

Riješite jednačbu $x^2 + ax + 2 = 0$ po varijabli x (tj. x je nepoznanica).

```
(%i54) solve(x^2 + a*x + 2 = 0, x);
```

```
(%o54) [x = - $\frac{\sqrt{a^2 - 8} + a}{2}$ , x =  $\frac{\sqrt{a^2 - 8} - a}{2}$ ]
```

4.6 Matrična jednačba

Koristimo funkciju `linsolve_by_lu`

```
(%i55) a: matrix([1, 2], [2, 1]);  
      b: [-1, 1];  
      linsolve_by_lu(a, b);  
      x: %[1];
```

```
(%o55)  $\begin{pmatrix} 1 & 2 \\ 2 & 1 \end{pmatrix}$ 
```

```
(%o56) [-1, 1]
```

```
(%o57) [ $\begin{pmatrix} 1 \\ -1 \end{pmatrix}$ , false]
```

```
(%o58)  $\begin{pmatrix} 1 \\ -1 \end{pmatrix}$ 
```

Rješenje možemo provjeriti na dva načina:

```
(%i59) a.x;  
      a.x - transpose(b);
```

```
(%o59)  $\begin{pmatrix} -1 \\ 1 \end{pmatrix}$ 
```

```
(%o60)  $\begin{pmatrix} 0 \\ 0 \end{pmatrix}$ 
```

5 Kraj

Na kraju je dobro obrisati sve varijable.

```
(%i61) kill(all);
```

```
(%o0) done
```


6 Linkovi

Opsežne manuale za Maximu možete naći ovdje:

<http://maxima.sourceforge.net/docs/manual/en/maxima.html>

<http://www.csulb.edu/~woollett/>

Upute za razne funkcije, pogotovo one kojih nema u helpu, možete pronaći pomoću Googlea. Npr:

`maxima linsolve.by.lu`

će vas odvesti na stranicu <http://cadadr.org/maxima/faq.html>

Usporedba Maxime, Maplea i Mathematice:

<http://beige.ucs.indiana.edu/P573/node35.html>